# $Id$
# Maintainer:
# Contributor: nut543 <kfs1@online.no>
# Contributor: Dale Blount <dale@archlinux.org>

pkgname=re2c
pkgver=1.0.2
pkgrel=1
arch=(i686 x86_64)
depends=(gcc-libs)
pkgdesc='A tool for generating C-based recognizers from regular expressions'
url='http://re2c.org'
license=(GPL)
source=(re2c-$pkgver.tar.gz::https://github.com/skvadrik/re2c/archive/$pkgver.tar.gz)
sha1sums=('d01b4d3c353bb32d5e4a1a61a377e0f4f03d4741')

build() {
  cd $pkgname-$pkgver/re2c
  ./autogen.sh
  ./configure --prefix=/usr
  make
}

check() {
  cd $pkgname-$pkgver/re2c
  make tests
}

package() {
  cd $pkgname-$pkgver/re2c
  make DESTDIR="$pkgdir" install
}
